/*The SAS program (HEI-2010 Individual Scores per day using NHANES 2009-2010 data (and FPED))
PerDay.SAS*/
/*INSTRUCTIONS – complete tasks 1-4 in this section, and run these SAS codes before proceeding to the HEI-2010 scoring program that follows*/

/*1. Create a folder on your computer “home folder”, and save the FPED data, NHANES data, Demographic data, and the required HEI-2010 macros in it. Specify the path to the folder. */

%let home = C:\Users\Documents\FPED_NHANES; /*In this Example, the “home” folder is in C Drive, within Documents, and is called FPED_NHANES. */

/*2. Libnames here specify the input files. */
libname NH “&home\NH”; 
libname FPED “&home\FPED”; /*In this Example, the FPED data are in a folder called “FPED”, and the NHANES and Demographic data are in a folder called “NH, all within the “home” folder. These are SAS datasets. */
/*3. Create a folder in the "Home" folder, where the output file, containing HEI-2010 component and total scores for each respondent, for the intake day, are to be exported. Specify the name of the folder. */

filename RES “&home\RES”; /*In this Example, the folder is called “RES”, within the “home” folder. The exported results will be a csv file called “hei2010r”. */
/*4. Read in required HEI-2010 scoring macros. These macros must be saved within the home folder. */
%include “&home\hei2010.beanspeas.allocation.macro.sas”;

%include “&home\hei2010.score.macro.sas”;

/*NOTE: Once you have completed all the steps above, all you need to do is run the SAS program below. Unless you used different names for your datasets and folders, no other action is required from you. */
title 'HEI-2010 scores for NHANES 2009-2010 day 1, AGE >= 2, RELIABLE DIETS, Include Pregnant and Lactating Women';

/*Step 1: locate the required datasets and variables */

*part a: get FPED data per day;
data FPED;

 set FPED.fped_dr1tot_0910;

run;
*part b: get individual total nutrient intake if reliable recall status;
data NUTRIENT (keep=SEQN WTDRD1 DR1TKCAL DR1TSFAT DR1TALCO DR1TSODI DR1DRSTZ DR1TMFAT DR1TPFAT);

  set NH.DR1TOT_F;

  if DR1DRSTZ=1; /*reliable dietary recall status*/
run;

*part c: get demographic data for persons aged two and older;
data DEMO (keep=SEQN RIDAGEYR RIAGENDR SDDSRVYR SDMVPSU SDMVSTRA);

  set NH.DEMO_F;

  if RIDAGEYR >= 2;

run;

/*Step 2: Combine the required datasets*/

proc sort data=FPED;

  by SEQN;

run;

proc sort data=NUTRIENT;

  by SEQN;

run;

proc sort data=DEMO;

  by SEQN;

run;

data COHORT;

  merge NUTRIENT (in=N) DEMO (in=D) FPED;

  by SEQN;

  if N and D;

run;

/*Step 3: Creates additional required variables: WHOLEFRT, MONOPOLY, ALLMEAT, SEAPLANT, ADDSUGC, SOLFATC, EXALCCAL, and EMPTYCAL10. */
data COHORT;

  set COHORT;

  by SEQN;

  WHOLEFRT=DR1T_F_CITMLB+DR1T_F_OTHER;
  MONOPOLY=DR1TMFAT+DR1TPFAT;

  ALLMEAT=DR1T_PF_MPS_TOTAL+DR1T_PF_EGGS+DR1T_PF_NUTSDS+DR1T_PF_SOY; 

  SEAPLANT=DR1T_PF_SEAFD_HI+DR1T_PF_SEAFD_LOW+DR1T_PF_NUTSDS+DR1T_PF_SOY;

 /**Calculate intake of Empty Calories (EMPTYCAL10)**/
       ADDSUGC=16*DR1T_ADD_SUGARS;

/*calories from added sugars*/

       SOLFATC=DR1T_SOLID_FATS*9;
/*calories from solid fat*/

      MAXALCGR=13*(DR1TKCAL/1000);  /*max grams of alcohol based on kcal intake*/
      if DR1TALCO <= MAXALCGR then EXALCCAL=0;  /*consumed less than max*/
        else if DR1TALCO > MAXALCGR then EXALCCAL=7*( DR1TALCO - MAXALCGR);  /*get cal from extra alc grams*/
      EMPTYCAL10=ADDSUGC+SOLFATC+EXALCCAL; /*total empty calories in hei2010 definition*/
run; 

/*Step 4: Run the macro to properly allocate the intakes of the FPED variable Legumes in cup equivalents (DR1T_V_LEGUMES) to either Total Protein Foods and Seafood and Plant Proteins (ALLMEAT and SEAPLANT) or Total Vegetables and Greens and Beans (DR1T_V_TOTAL and DR1T_V_DRKGR).  The four resulting variables from this step, named legume_added_DR1T_V_TOTAL, legume_added_BEANGRN, legume_added_ALLMEAT, and legume_added_SEAPLANT are all used in the HEI scoring macro in Step 5*/
%LEG2010A (indat= COHORT,

           kcal= DR1TKCAL,

           allmeat=ALLMEAT,


   seaplant=SEAPLANT,


   v_total=DR1T_V_TOTAL,


   v_drkgr= DR1T_V_DRKGR,


   legumes= DR1T_V_LEGUMES,


   outdat=AFTERLEG);

/*Step 5: Apply the HEI-2010 scoring macro. */
%HEI2010 (indat= AFTERLEG,

          kcal= DR1TKCAL,


  lv_total= legume_added_DR1T_V_TOTAL,


  lbeangrn= legume_added_BEANGRN,


  f_total= DR1T_F_TOTAL,


  wholefrt=WHOLEFRT,


  g_whl= DR1T_G_WHOLE,


  d_total= DR1T_D_TOTAL,

       lallmeat= legume_added_ALLMEAT,


  lseaplant= legume_added_SEAPLANT,


  monopoly=MONOPOLY,


  sfat=DR1TSFAT,


  sodi=DR1TSODI,


  g_nwhl= DR1T_G_REFINED,


  Emptycal10=EMPTYCAL10,


  outdat=HEI2010);

/*Step 6: Displays and saves the results. */ 

*part a: this program saves one HEI-2010 score for each individual, based on one 24HR;

data HEI2010R (keep=SEQN DR1TKCAL HEIX1_TOTALVEG HEIX2_GREEN_AND_BEAN HEIX3_TOTALFRUIT

     HEIX4_WHOLEFRUIT HEIX5_WHOLEGRAIN HEIX6_TOTALDAIRY HEIX7_TOTPROT HEIX8_SEAPLANT_PROT

     HEIX9_FATTYACID HEIX10_SODIUM HEIX11_REFINEDGRAIN HEIX12_SOFAAS

     HEI2010_TOTAL_SCORE);

  set HEI2010;

  run;

*part b: calculates an unweighted mean across all individuals in group;
proc means n nmiss min max mean data=HEI2010R;

run;

*part c: saves results as CSV file one line per day;

proc export data=HEI2010R
  file=RES
  dbms=csv

  replace;

run;

